A COMPUTER PROGRAM FOR CONVERTING RECTANGULAR COORDINATES TO LATITUDE-LONGITUDE COORDINATES

By A.T. Rutledge

U.S. GEOLOGICAL SURVEY

Water-Resources Investigations Report 89-4070

Prepared in cooperation with the

NORTHWEST FLORIDA WATER MANAGEMENT DISTRICT
ST. JOHNS RIVER WATER MANAGEMENT DISTRICT
SOUTH FLORIDA WATER MANAGEMENT DISTRICT
SOUTHWEST FLORIDA WATER MANAGEMENT DISTRICT
SUWANNEE RIVER WATER MANAGEMENT DISTRICT

Tallahassee, Florida

1989

DEPARTMENT OF THE INTERIOR
MANUEL LUJAN, JR., Secretary

U.S. GEOLOGICAL SURVEY
Dallas L. Peck, Director

For additional information
write to:

District Chief
U.S. Geological Survey
Suite 3015
227 North Bronough Street
Tallahassee, Florida 32301

Copies of this report can
be purchased from:

U.S. Geological Survey
Books and Open-File Reports
Federal Center, Building 810
Box 25425
Denver, Colorado 80225

CONTENTS

ILLUSTRATIONS

A COMPUTER PROGRAM FOR CONVERTING RECTANGULAR COORDINATES TO LATITUDE-LONGITUDE COORDINATES

By A.T. Rutledge

ABSTRACT

A Fortran-77 computer program was developed for converting the coordinates of any rectangular grid on a map to coordinates on a grid that is parallel to lines of equal latitude and longitude. Using this program in conjunction with ground-water flow models, the user can extract data and results from models with varying grid orientations and place these data into a grid structure that is oriented parallel to lines of equal latitude and longitude. All cells in the rectangular grid must have equal dimensions, and all cells in the latitude-longitude grid must measure 1 minute by 1 minute. This program is applicable if the map used shows lines of equal latitude as arcs and lines of equal longitude as straight lines and assumes that the Earth's surface can be approximated as a sphere.

The program user enters the row number, column number, and latitude and longitude of the midpoint of the cell for three test cells on the rectangular grid. The latitude and longitude of boundaries of the rectangular grid also are entered. By solving sets of simultaneous linear equations, the program calculates coefficients that are used for making the conversions.

As an option in the program, the user may build a ground-water model data file based on a grid that is parallel to lines of equal latitude and longitude. The program reads a data file based on the rectangular coordinates and automatically forms the new data file.

INTRODUCTION

Finite-difference grids that are used in conjunction with ground-water flow models commonly are not oriented along latitude or longitude lines but are oriented in a way that enables the modeler to show predominant flow paths along one axis of the grid. Extracting data or results from a model for other related uses, including modeling at a detailed level, may be complicated because the model grid does not coincide with a universal land coordinate system.

The Fortran-77 computer program described here allows the user to transfer data from a rectangular grid that can be oriented in any direction on a map (referred to as the old grid) to a grid that is oriented such that it consists of lines of equal latitude and longitude (referred to as the new grid). Cells, which are the elemental units of model grids defined by two coordinates, must be rectangles of uniform dimensions on the old grid and must be areas measuring 1 minute of latitude by 1 minute of longitude on the new grid. New grid coordinates are assigned such that the origin is at the northwest corner and row number increases to the south, whereas column number increases to the east. The size of cells on the old grid may be different from that of the new grid, but the results may not be useful if old grid cells are substantially smaller than new grid cells because the coordinates of several old grid cells might translate to the coordinates of a single cell in the new grid.

CONVERSION OF RECTANGULAR COORDINATES TO LATITUDE-LONGITUDE COORDINATES

The simplified conversion of old grid coordinates to new grid coordinates is shown in figure 1. In this example, both grids are rectangular. Simple mathematical relations can be derived to calculate the coordinates on the new grid at a location that is defined by coordinates on the old grid. This is shown as equations (1) and (2) in figure 1. The new row number (eq. 1) is calculated through the use of a first-order polynomial in two variables: the old row number and the old column number. The new column number (eq. 2) is determined similarly. These two equations have their coefficients (X1-X6) determined by the solution of three simultaneous first-order equations written for three test cells. New grid test cells are chosen such that their midpoints coincide as closely as possible with the midpoints of old grid cells. After the six coefficients of equations (1) and (2) have been determined, the equations will give new grid coordinates of the midpoints of any old grid cell. Ground-water flow-model variables, such as transmissivity, assigned to an old grid cell then may be assigned to that cell's corresponding cell on the new grid.

The intended use of this program is for converting coordinates on the old grid to coordinates on the new grid, where the new grid is made up of lines that are parallel to lines of equal latitude and equal longitude and where lines of equal latitude are arcs and lines of equal longitude are straight lines (as is the case on a map that is a Lambert conformal conic projection). Because the new grid coordinate system therefore is radial, instead of rectangular, adjustments in the simplified grid conversion in figure 1 are necessary.

The necessary adjustments to the previously described approach for determining a new grid column number are shown in figure 2. The diagram on the left portrays two equal east-west displacements at two different latitudes. Because the lines of equal longitude converge, the equal displacements cause the two new column numbers to differ. The diagram on the right shows the derivation of an expression for the adjustment necessary, and the revised equation for J_{new} is shown (eq. 3). It is assumed here that the Earth's surface can be approximated as a sphere. Before equation 3 is solved, an expression for approximate latitude must be solved. The error associated with this calculation (eq. 4) is relatively small and considered insignificant because approximate latitude is used only for determining cosine of latitude, and errors transferred by way of the cosine function are small.

The need for an adjustment in the equation for a new grid row number is illustrated by the left diagram in figure 3. The previous equations (fig. 1) would be inadequate because they do not allow for the apparent northward curvature of the arcs of equal latitude. The adjustment for this curvature is represented by Δ I in equation 5. The adjustment is formulated to be zero at the western edge of the new grid and to increase eastward. The derivation of the expression for this adjustment (eq. 6) is shown on the right diagram in figure 3. An expression for the R term is shown in figure 4 (eq. 7).

The procedures used by the program are illustrated in figure 5. User input consists of latitude-longitude coordinates of the boundaries of the new grid and of three test cells. Equation 7 is executed first, and R, a characteristic of the new grid, is determined. By use of location data pertaining to the three test cells, the program solves for the coefficients of equations 4, 3, and 5, in that order. On the basis of the same equations, having their coefficients as known values, the program then converts old grid coordinates to new grid coordinates.

USER PROCEDURE FOR COMPUTER PROGRAM APPLICATION

To execute this program, the user needs the latitude and longitude of the northwest and the southeast corners of the new grid, the coordinates of three test cells on the old grid, and the latitude and longitude of the midpoint of these cells. The program prompts user input. The test cells should be located as far from each other as possible and, preferably, should be cells that represent actively modeled areas. Any two test cells should not be located at the same approximate latitude or the same approximate longitude. After the program calculates conversion coefficients, the user may test the program by requesting new-grid coordinates of the test cells previously given. The calculated coordinates of these should be equal to the values given during input. When the user is satisfied that data have been entered correctly, the program can be tested further by requesting the new grid coordinates of cells not previously given.

The user may exit the "cell testing" procedure and then choose to either terminate the program execution or opt to build a data file in new grid coordinates. If the second option is chosen, a file named "OLDDATA" must be in the directory. This is a data file configured on the old grid format. It can be transmissivity, leakance, head, or any other model input data. The program prompts the user to enter the dimensions of the old grid (number of rows, number of columns) and then creates a data file named "NEWDATA," which is configured on latitude-longitude



$$I_{new} = (I_{old} \cdot X1) + (J_{old} \cdot X2) + X3 \qquad (1)$$

$$J_{new} = (I_{old} \cdot X4) + (J_{old} \cdot X5) + X6 \qquad (2)$$

EXPLANATION

(I_{old}, J_{old}) ROW, COLUMN COODINATES ON OLD GRID

(I_{new}, J_{new}) ROW, COLUMN COORDINATES ON NEW GRID

X1, X2, X3, X4, X5, X6 COEFFICIENTS OF THE TWO EQUATIONS

Figure 1.––Simplified grid conversion based on translation between two rectangular coordinate systems.



$$J_{new} = \frac{(I_{old} \cdot X4) + (J_{old} \cdot X5) + X6}{W},$$

where: W = width of one new grid column (function of latitude).

To derive expression for W :



r = distance to the axis of the Earth
Lat = latitude
E = radius of the Earth

$$W = \frac{2r\pi}{360 \cdot 60} \text{ and } r = E \cdot \mathrm{Cos}(Lat), \text{ so}$$

$$W = \frac{2E\pi}{360 \cdot 60} \cdot \mathrm{Cos}(Lat).$$

The term $\frac{2E\pi}{360 \cdot 60}$ is constant, so it may be deleted from the equation for J_{new} before solving for X4, X5, and X6 :

$$J_{new} = \frac{(I_{old} \cdot X4) + (J_{old} \cdot X5) + X6}{\mathrm{Cos}\ (Lat)}, \qquad (3)$$

where $Lat = (I_{old} \cdot X1) + (J_{old} \cdot X2) + X3$, approximately. (4)

Figure 2.—Adjustments necessary for calculating new grid column number.



$$I_{new} = (I_{old} \cdot X7) + (J_{old} \cdot X8) + X9 + \Delta I, \quad (5)$$

where ΔI = upward displacement of a new grid row with respect to the old (rectangular) grid. ΔI is a function of longitude (or new grid column number).

To derive an expression for ΔI :



From the Pythagorean Theorom,

$$\Delta I = R - \sqrt{R^2 - D^2} .$$

To express D in the same units as ΔI (minutes of latitude),

$$D = J_{new} \cdot \text{Cos (Lat)}, \text{ so}$$

$$\Delta I = R - \sqrt{R^2 - (J_{new} \cdot \text{Cos (Lat)})^2} , \quad (6)$$

and R is expressed in minutes of latitude.

Figure 3.—–Adjustments necessary for calculating new grid row number.



R = average radius of circle representing an arç of equal latitude.

$$\frac{R - \frac{c}{2}}{b} = \frac{R + \frac{c}{2}}{a}$$

$$aR - \frac{ac}{2} = bR + \frac{bc}{2}$$

$$R\ (a - b) = \frac{c}{2}\ (a + b)$$

$$R = \frac{c}{2}\ \frac{(a + b)}{(a - b)}.$$

Lat_N and Lat_S are latitudes of northern and southern boundaries, in degrees, and $Long_E$ and $Long_W$ are eastern and western boundaries, in degrees.

To express a and b in units of length, consider figure 2 :

$$a = \frac{(Long_W - Long_E)}{360}\ 2\pi E\ \mathrm{Cos}(Lat_S),$$

$$b = \frac{(Long_W - Long_E)}{360}\ 2\pi E\ \mathrm{Cos}(Lat_N)\cdot$$

Therefore :
$$\frac{a + b}{a - b} = \frac{\mathrm{Cos}(Lat_S) + \mathrm{Cos}(Lat_N)}{\mathrm{Cos}(Lat_S) - \mathrm{Cos}(Lat_N)},\ \text{dimensionless},$$

and, expressing R and C in minutes of latitude:

$$R = \frac{c}{2}\ \frac{(a + b)}{(a - b)} = \frac{60\ (Lat_N - Lat_S)}{2} \left[\frac{\mathrm{Cos}(Lat_S) + \mathrm{Cos}(Lat_N)}{\mathrm{Cos}(Lat_S) - \mathrm{Cos}(Lat_N)}\right],\ \text{or}$$

$$R = 30\ (Lat_N - Lat_S) \left[\frac{\mathrm{Cos}(Lat_S) + \mathrm{Cos}(Lat_N)}{\mathrm{Cos}(Lat_S) - \mathrm{Cos}(Lat_N)}\right]. \qquad (7)$$

Figure 4.––Derivation of equation for R.



Figure 5.--Flow chart of procedures used by the program.

coordinates. The only cells in the file "NEWDATA" that are given nonzero values are those that correspond to midpoints of the old grid cells that have nonzero values. It is left to the user to develop methods (interpolation, weighted averaging, and so forth) for assigning values to other cells based on needs associated with a particular modeling effort. Methods for this purpose were developed by Luckey and Ferrigno (1982).

A listing of the Fortran-77 code of program "GRID" is in the back of this report. Modification may be necessary to allow the user to vary the format of the data files "OLD DATA" and "NEW-DATA" or the dimensions of the data arrays that are stored on these files ("DATA1" on "OLDDATA" and "DATA2" on "NEWDATA"). Lines that may need modification are indicated with an asterisk on the right.

SELECTED REFERENCE

Luckey, R.R. and Ferrigno, C.R., 1982, A data-management system for areal interpretive data for the high plains in parts of Colorado, Kansas, Nebraska, New Mexico, Oklahoma, South Dakota, Texas, and Wyoming: U.S. Geological Survey Water-Resources Investigations Report 82-4072, 112 pages.

APPENDIX

page 11 follows

APPENDIX--Computer program listing

```
      PROGRAM GRID
C
C--------THIS PROGRAM ENABLES THE USER TO TRANSLATE THE COORDINATES ----------
C--------OF CELLS ON A RECTANGULAR CARTESIAN GRID (OLD) TO COORDINATES---------
C--------BASED ON  LATITUDE AND LONGITUDE (NEW). ---------------------------
C--------THE RECTANGULAR CARTESIAN GRID NEED NOT BE ALIGNED WITH ANY----------
C--------PART OF THE LAT-LONG GRID. THE CONVERSION EQUATIONS THUS DERIVED------
C--------ARE USED TO TRANSLATE GROUND-WATER MODEL DATA FROM THE OLD-----------
C--------GRID FORMAT (FILE "OLDDATA") TO THE NEW GRID FORMAT (FILE "NEWDATA")--
C--------IF THE USER CHOOSES. ------------------------------------------------
C
       IMPLICIT REAL (A-Z)
       INTEGER N, S, E, W, I, J, ROWSO, COLSO, ROWSN, COLSN, NUM
       DIMENSION DATA1(100,100)                                          *
       DIMENSION DATA2(800,800)                                          *
       COMMON // DATA2
       OPEN (11, FILE='OLDDATA', STATUS='OLD')                           *
       OPEN (12, FILE='NEWDATA', STATUS='NEW', ACCESS='DIRECT',          *
      % FORM='FORMATTED', RECL=8000)                                     *
C
C----------OBTAIN USER INPUT REGARDING NEW GRID----------
C
       PRINT *, '     WHERE IS THE SE CORNER OF THE NEW GRID?   '
       PRINT *, '           ENTER THE FOLLOWING:'
       PRINT *,'LAT DEGREES, LAT MINUTES, LONG DEGREES, LONG MINUTES <RET
      &> '
       READ (*,*) LATSEDG, LATSEMN, LNGSEDG, LNGSEMN
       PRINT *, ' '
       PRINT *, '     WHERE IS THE NW CORNER OF THE NEW GRID?'
       PRINT *,'          ENTER THE FOLLOWING '
       PRINT *,'LAT DEGREES, LAT MINUTES, LONG DEGREES, LONG MINUTES <RET
      &> '
       READ (*,*) LATNWDG, LATNWMN, LNGNWDG, LNGNWMN
       PRINT *, ' '
C
C ----------CALCULATE PROPERTIES OF NEW GRID-------------------
C
       LATN= LATNWDG + (LATNWMN/60)
       LATS= LATSEDG + (LATSEMN/60)
       COSN= COS(LATN*0.01745)
       COSS= COS(LATS*0.01745)
       R= (30*(LATN-LATS)*(COSS+COSN))/(COSS-COSN)
       PRINT *, ' '
       WRITE (*,*) 'R=' , R , '  (UNITS ARE MINUTES OF LATITUDE)'
       PRINT *, ' '
       N=(LATNWDG*60)+LATNWMN
       S=(LATSEDG*60)+LATSEMN
       E=(LNGSEDG*60)+LNGSEMN
       W=(LNGNWDG*60)+LNGNWMN
```

```
C
C-----------------OBTAIN USER INPUT DATA REGARDING 3 TEST CELLS--------------
C-----------------ON OLD GRID FOR WHICH LAT AND LONG ARE KNOWN-----------------
-
C
   50 PRINT *,'NOW ENTER COORDINATES OF 3 TEST CELLS SPREAD OVER AREA:'
      PRINT *, ' '
      PRINT *,' FIRST CELL---- ENTER:'
      PRINT *,'LAT DEGREES, LAT MINUTES, LONG DEGREES, LONG MINUTES, OLD
     & ROW NUMBER, OLD COLUMN NUMBER, <RET>'
      READ (*,*) LTDG1, LTMN1, LGDG1, LGMN1, IO1, JO1
      PRINT *, ' '
      PRINT *,' SECOND CELL---- ENTER:'
      PRINT *,'LAT DEGREES, LAT MINUTES, LONG DEGREES, LONG MINUTES, OLD
     & ROW NUMBER, OLD COLUMN NUMBER, <RET>'
      READ (*,*) LTDG2, LTMN2, LGDG2, LGMN2, IO2, JO2
      PRINT *, ' '
      PRINT *,' THIRD CELL---- ENTER:'
      PRINT *,'LAT DEGREES, LAT MINUTES, LONG DEGREES, LONG MINUTES, OLD
     & ROW NUMBER, OLD COLUMN NUMBER, <RET>'
      READ (*,*) LTDG3, LTMN3, LGDG3, LGMN3, IO3, JO3
C
C --------CALCULATE PROPERTIES OF 3 TEST CELLS: -----------------
C
      IN1=N-((LTDG1*60)+LTMN1)
      JN1=W-((LGDG1*60)+LGMN1)
      IN2=N-((LTDG2*60)+LTMN2)
      JN2=W-((LGDG2*60)+LGMN2)
      IN3=N-((LTDG3*60)+LTMN3)
      JN3=W-((LGDG3*60)+LGMN3)
      LAT1= LTDG1 + (LTMN1/60)
      LAT2= LTDG2 + (LTMN2/60)
      LAT3= LTDG3 + (LTMN3/60)
      COSLAT1= COS(LAT1*0.01745)
      COSLAT2= COS(LAT2*0.01745)
      COSLAT3= COS(LAT3*0.01745)
C
C --------------------- DETERMINE X1, X2, AND X3 -----------------
C
      CALL SOLVE (IO1, JO1, 1.0, IO2, JO2, 1.0, IO3, JO3, 1.0, LAT1,
     $ LAT2, LAT3, X1, X2, X3)
C
C  -------------------DETERMINE X4, X5, AND X6---------------------
C
      CALL SOLVE (IO1/COSLAT1, JO1/COSLAT1, 1/COSLAT1, IO2/COSLAT2,
     $ JO2/COSLAT2, 1/COSLAT2, IO3/COSLAT3, JO3/COSLAT3, 1/COSLAT3,
     $ JN1, JN2, JN3, X4, X5, X6)
C
C ---------DETERMINE "DELTA-I" VALUES FOR THE THREE TEST CELLS-------
C
      DELI1= R-((R**2)-(JN1*COS(LAT1*0.01745))**2)**0.5
      DELI2= R-((R**2)-(JN2*COS(LAT2*0.01745))**2)**0.5
      DELI3= R-((R**2)-(JN3*COS(LAT3*0.01745))**2)**0.5
```

```
C
C ----------------DETERMINE X7, X8, AND X9: -----------------------
C
      CALL SOLVE (I01, J01, 1.0, I02, J02, 1.0, I03, J03, 1.0,
     $ IN1-DELI1, IN2-DELI2, IN3-DELI3, X7, X8, X9)
C
C ----CALCULATE NEW COORDINATES OF SELECTED CELLS ON OLD GRID----
      PRINT *, ' '
      PRINT *, '   THE NINE COEFFICIENTS DERIVED FOR THIS COORDINATE TRA
     &NSFORMATION ARE THUS:'
  200 PRINT *,  '     X1           X2           X3           X4
     #X5          X6           X7           X8           X9'
      WRITE (*,250) X1, X2, X3, X4, X5, X6, X7, X8, X9
      PRINT *, ' '
      PRINT *, '         THE FOLLOWING PROCEDURE MAY BE USED TO TEST THE
     & COORDINATE TRANSFORMATION: '
  250 FORMAT (9F12.6)
  300 PRINT *, ' '
      PRINT *, ' '
      PRINT *, 'ENTER OLD ROW NUMBER <RET> OLD COLUMN NUMBER <RET>'
      PRINT *, '(TO EXIT THIS CELL-TESTING PROCEDURE, ENTER  "0" FOR ROW
     & NUMBER)'
      READ (*,*) IO
      IF(INT(IO).EQ.0) GOTO 400
      READ (*,*) JO
      LAT= (IO*X1) + (JO*X2) + (X3)
      JN= ((X4*IO)+(X5*JO)+X6)/(COS(LAT*0.01745))
      DELI= (R-((R**2)-(JN*COS(LAT*0.01745))**2)**0.5)
      IN= (X7*IO)+(X8*JO)+X9+DELI
      PRINT *, '      OLD       OLD          NEW       NEW '
      PRINT *, '      ROW      COLUMN        ROW      COLUMN
      WRITE (*,350) IO, JO, IN, JN
  350 FORMAT (4F10.0)
      GOTO 300
  400 CONTINUE
C
C          CHOOSE TO RUN AGAIN WITH NEW TEST CELLS, OR TO TRANSFORM A
C          DATA FILE, OR TO TERMINATE JOB.
C
      PRINT *, 'IF YOU WANT TO RUN AGAIN, USING DIFFERENT TEST CELLS,
     %ENTER THE NUMBER "1" '
      PRINT *, ' '
      PRINT *, 'IF GRID CONVERSION IS WORKING SUITABLY, AND YOU ARE READ
     $Y TO CONVERT A DATA ARRAY FROM OLD TO NEW GRID, THEN ENTER "2".'
      PRINT *, 'NOTE: THE DATA FILE TO BE TRANSFORMED MUST BE UNDER THIS
     $ DIRECTORY AND NAMED "OLDDATA". '
      PRINT *, ' '
      PRINT *, 'TO TERMINATE PROGRAM, ENTER ANY OTHER NUMBER'
      READ (*,*) NUM
      IF (NUM.EQ.1) GOTO 50
      IF(NUM.NE.2) GOTO 800
```

```
C
C ---------OBTAIN USER INPUT REGARDING OLD GRID --------------------
C
      PRINT *,' ENTER THE NUMBER OF ROWS IN OLD GRID:'
      READ (*,*) ROWSO
      PRINT *, 'ENTER THE NUMBER OF COLUMNS IN OLD GRID:'
      READ (*,*) COLSO
      ROWSN=N-S
      COLSN=W-E
C
C --------INITIALIZE DATA1, THEN READ DATA1 FROM FILE "OLDDATA"------
C
      DO 450 IO=1, ROWSO
      DO 450 JO=1, COLSO
  450 DATA1(INT(IO),INT(JO))=0.0
      DO 500 IO=1,ROWSO
  500 READ(11,550)(DATA1(INT(IO),INT(JO)), JO=1, COLSO)
  550 FORMAT (20F4.0)                                                  *
C
C-----------INITIALIZE AND DESIGNATE VALUES OF DATA2 FOR EACH ---------
C-----------CELL COINCIDENT WITH A MIDPOINT OF A CELL OF DATA1.---------
C
      DO 600 IN=1,ROWSN
      DO 600 JN=1,COLSN
  600 DATA2(INT(IN),INT(JN))=0.0
      DO 650 IO=1,ROWSO
      DO 650 JO=1,COLSO
      LAT= (IO*X1) + (JO*X2) + (X3)
      JN= ((X4*IO) + (X5*JO) + X6)/(COS(LAT*0.01745))
      DELI=(R-((R**2) - (JN*COS(LAT*0.01745))**2)**0.5)
      IN= (X7*IO) + (X8*JO) + X9 + DELI
  650 DATA2(INT(IN+0.5),INT(JN+0.5))= DATA1(INT(IO),INT(JO))
C
C --------WRITE MODEL DATA ONTO A LAT-LONG GRID DATA FILE------------
C ---------------(DATA2 WRITTEN ON FILE "NEWDATA") -------------------
C
      DO 700 IN=1, ROWSN
  700 WRITE (12,750)(DATA2(INT(IN+0.5),INT(JN+0.5)), JN=1,COLSN)
  750 FORMAT (510F10.4)
*
      CLOSE (12, STATUS= 'KEEP')
      GOTO 900
  800 CONTINUE
      CLOSE (12, STATUS= 'DELETE')
  900 CONTINUE
      CLOSE (11, STATUS= 'KEEP')
      STOP
      END
```

```
C         THIS ROUTINE PERFORMS GAUSSIAN ELIMINATION, WITH MAXIMUM
C         COLUMN PIVOTING, ON THE AUGMENTED 3*4 MATRIX "MAT". THE LEFT
C         3*3 SIDE OF THE MATRIX IS MADE UP OF THE VALUES OF "A" AND
C         THE RIGHT SIDE, BY THE VALUES OF "B". BACKWARD SUBSTITUTION
C         IS USED TO CALCULATE THE SOLUTION: THE VALUES OF "X".
C
      SUBROUTINE SOLVE (A11, A12, A13, A21, A22, A23, A31, A32, A33,
     & B1, B2, B3, X1, X2, X3)
      INTEGER P
      REAL MX
      REAL MAT(3,4)
      REAL M(3,3)
      DIMENSION XC(3)
      MAT(1,1)= A11
      MAT(1,2)= A12
      MAT(1,3)= A13
      MAT(2,1)= A21
      MAT(2,2)= A22
      MAT(2,3)= A23
      MAT(3,1)= A31
      MAT(3,2)= A32
      MAT(3,3)= A33
      MAT(1,4)= B1
      MAT(2,4)= B2
      MAT(3,4)= B3
C
C -------------------- GAUSSIAN ELIMINATION: --------------------
C
      DO 40 I=1,2
          P=I
          MX=MAT(I,I)
          DO 15 K=I,2
              IF (ABS(MX).LT.ABS(MAT(K+1,I))) THEN
                  P=1+K
                  MX= MAT(K+1,I)
              END IF
   15     CONTINUE
          IF (P.NE.I) THEN
              DO 18 K=1, 4
              TEMP=MAT(I,K)
              MAT(I,K)=MAT(P,K)
   18         MAT(P,K)= TEMP
          END IF
   19     DO 30 J=I+1,3
              M(J,I)=(MAT(J,I))/(MAT(I,I))
                  DO 20 K=1,4
   20                 MAT(J,K)= MAT(J,K)-(M(J,I)*MAT(I,K))
   30 CONTINUE
   40 CONTINUE
C
```

```
C ------------------- BACKWARD SUBSTITUTION: ----------------
C
      XC(3)=(MAT(3,4))/MAT(3,3)
      DO 70 I=2, 1, -1
          SUM=0.
          DO 60 J=I+1, 3
   60          SUM= SUM + MAT(I,J)*XC(J)
   70          XC(I)= (MAT(I,4)-SUM)/MAT(I,I)
      X1= XC(1)
      X2= XC(2)
      X3= XC(3)
      RETURN
      END
```

☆U.S. GOVERNMENT PRINTING OFFICE 1989 731 097/00010